
1 

UAb – 21010: Arquitetura de Computadores – eFólio-B 2022 comentários e resolução 

E-fólio B, notas lançadas  

por José Coelho - Terça, 18 Janeiro 2022, 11:47  

  

  Caros estudantes, 

Estão lançadas as notas do e-fólio B. A distribuição dos estudantes pelas notas foi a seguinte: 

E-fólio B E-fólios: 0-1 1-2 2-3 3-4 4-5 5-6 6-7 7-8 Total Geral 

(em branco)  250 23 19 5 2   299

0-0,5   1 1 1 1   4

0,5-1    3 3    6

1-1,5     3 8   11

1,5-2      6 4   10

2-2,5     2 3 3 3 11

2,5-3       2   2

3-3,5         1 1

3,5-4         5 5

Total Geral  250 24 23 14 20 9 3 6 349

 

Podemos ver que nos 50 trabalhos entregues no e-fólio B, temos 6 notas com mais de 3 valores, que 

ficam no total dos e-fólios com nota acima de 7 valores. Como sabem, podem ir a e-fólio global, 

todos os que tenham 3.5 valores ou mais, no total dos e-fólios. Estão 44 estudantes em condições 

de ir ao e-fólio Global. Embora do ponto de vista do número de estudantes inscritos este valor é 

reduzido, tendo em atenção o número de trabalhos entregues do e-fólio B, considero que este valor 

é bastante positivo. Infelizmente não houve a mesma participação neste e-fólio que no e-fólio A, 

mas os estudantes que não tiveram disponibilidade para este e-fólio, podem no caso de terem 

menos de 3,5 valores no total dos e-fólios, procurar fazer o exame na época de recurso. Nesta UC, 

quem tenha feito ou faça até à data da prova, as atividades formativas, que naturalmente não é 

pouco já que são em grande número, tem boas possibilidades de ficar com a prática necessária para 

obter a aprovação via exame. 

Vejam os comentários do feedback individual, e se tiverem alguma questão coloquem. A fórmula 

de calculo é idêntica ao e-fólio A. 

Em termos de critérios, convém clarificar a cotação atribuída às alíneas: 

 0,1 – apenas iniciado, vários problemas;  

 0,5 – está perto de ter a alínea completa (contém bugs ou falta muito pouco). 

Os resultados corretos são os seguintes: 

8000 : 0001 0002 0003 0004 0005 0006 0007 0008 ........ 

8008 : 0009 000a 000b 000c 000d 000e 000f 0010 ........ 

8010 : 0000 0001 0002 0001 0002 0003 0004 0000 ........ 

8018 : 0007 0005 0002 0003 0004 0001 0004 0005 ........ 

8020 : 0001 0000 0001 0000 0000 0000 0001 0000 ........ 

8028 : 0000 0000 0000 0000 0000 0000 0001 0000 ........ 

8030 : 0001 0001 0001 0000 0001 0000 0000 0001 ........ 

8038 : 0001 0000 0000 0000 0000 0000 0001 0000 ........ 

8040 : 0001 0001 0001 0001 0000 0001 0000 0000 ........ 

8048 : 0001 0001 0000 0000 0000 0000 0000 0001 ........ 

8050 : 0000 0001 0000 0000 0000 0000 0000 0000 ........ 

8058 : 0001 0000 0000 0000 0000 0001 0000 0000 ........ 

8060 : 0000 0001 0000 0001 0000 0000 0000 0000 ........ 

8068 : 0000 0000 0000 0000 0000 0000 0000 0001 ........ 

8070 : 0000 0000 0000 0000 0000 0000 0000 0000 ........ 

8078 : 0000 0001 0000 0000 0000 0000 0000 0001 ........ 

8080 : 832f 2102 800a 8200 0022 2001 0008 2002 ...."... 

8088 : 8202 8000 0000 0200 0020 0000 0003 8200 ........ 

8090 : 0002 2002 0001 2000 0000 0000 0100 0002 ........ 

8098 : 0200 0000 0000 8100 0002 0000 0008 2002 ........ 

80a0 : 0002 0000 0002 8200 0000 0000 0000 0200 ........ 

80a8 : 0000 0000 0008 0000 8000 0000 0002 0000 ........ 

80b0 : 8000 0002 0000 0000 0100 0000 8000 2000 ........ 

80b8 : 0002 0000 0000 0000 0021 0000 0002 0200 ....!... 

Podem com base na primeira diferença, localizar o problema no vosso código, se for caso disso. 

Em termos de instruções e ciclos de relógio, acabou-se por não dar muito relevo à eficiência, 

atendendo a que houve apenas 3 estudantes a ter a funcionalidade completa, e com alíneas parciais 

torna-se complicado utilizar esse indicador. Mas fica aqui o registo do top 3 dos programas mais 

eficientes: 

1. Pedro Santos, com 61 266 824 ciclos, 5 273 260 instruções 

2. Daniel Oliveira, com 118 618 029 ciclos, 8 278 120 instruções 

3. Joel José Ginga, com 331 703 901 ciclos, 33 998 891 instruções 

Durante o e-fólio houve uma questão sobre a relação deste e-fólio com as atividades formativas. De 

facto o e-fólio foi inspirado numa delas: algoritmo para calcular os números primos, exercício 10. O 

conjunto dos números primos pode ser definido da mesma forma, um conjunto com definição 

recursiva. As várias possibilidades de implementação deste tipo de conjunto, são idênticas. Seria 

mais interessante ter pedido este e-fólio sobre os conjuntos dos números primos, mas infelizmente 

já existia a atividade formativa, claro que há várias maneiras de se calcular números primos, mas 

ficaria um e-fólio demasiado colado a uma atividade formativa.  

Em vez de listar os erros cometidos, que não tenho a certeza de chegar a quem precise, é talvez 

preferível dar uma resolução comentada.  

Alínea A: 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

                ; função TesteNM solicitada na alínea A 

                ; Entrada: R1 - valor de N; R2 - valor de M 

                ; Saída: R1 - resultado 

                ; Registos: R1 - R4 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

TesteNM:        MOV R3, R1    ; R3=N é preciso manter para ver se as expressões 

são iguais 

NMcalcNmodM:    DIV R1, R2    ; R2=N%M; R1-livre.  

                              ; Esta sub-expressão é utilizada nas outras 

expressões,  

                              ; não sendo necessário mais M, pelo que o valor é 

perdido 

                              ; Notar ainda que se R2 é 0, o resultado após o 

DIV em R2 é 0, conforme pretendido 

NMpasso2:       MOV R1, R2    ; R1 com cópia de N%M para a segunda expressão 

                SHL R2,1      ; teste N==2*(N%M)+1 primeiro, já que é a 

expressão mais curta 

                INC R2         

                CMP R2, R3     

                BR.Z NMsucesso 

                DEC R2        ; teste N=(N%M)^2+2*(N%M) - R2=2*(N%M) 

                MOV R4, R1    ; tem de se calcular o quadrado de N%M 

                MUL R4, R1    ; R1=(N%M)^2; R4 - livre 

                ADD R2, R1    ; R2 - expressão completa 

                CMP R2, R3 

                BR.NZ NMfalha 

NMsucesso:      MOV R1, 1 

                RET 

NMfalha:        MOV R1, R0 

                RET 

Alínea B 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

                ; função TesteN solicitada na alínea B 

                ; Entrada: R1 - valor de N 

                ; Saída: R1 - resultado 

                ; Registos: R1, R2 

                ; reutiliza os labels NMsucesso e NMfalha 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

TesteN:         CMP R1, 2       ; R1=N mantém o valor 

                BR.NP NMsucesso   

                                ; M tem de ser iterado, é preferivel de 0 a N-1 

                MOV R2, R0      ; R2=M 

Nciclo:         CMP R2, R1 

                BR.Z NMfalha    ; todos os valores de M testados, retornar falha 

                 

                ; verificar se M pertence a S através da chamada recursiva 

                PUSH R1         ; guardar os dois registos utilizados 

                PUSH R2 

                MOV R1, R2      ; colocar o argumento 

                CALL TesteN 

                CMP R1, R0      ; valor de retorno, teste para ficar com o 

resultado nas flags 

                POP R2          ; estas operações de reposição de registos, não 

alteram as flags 

                POP R1 

                BR.Z Nprox      ; retorno falso, pelo que não se testa este M 

                ; chamar TesteNM com os valores atuais 

                PUSH R1         ; não é necessário guardar R3 e R4, podem ser 

destruidos por TesteNM 

                PUSH R2 

                CALL TesteNM     

                CMP R1, R0      ; R1 fica com o resultado, se !=0 é positivo 

                POP R2 

                POP R1 

                BR.NZ NMsucesso ; caso seja positivo, retornar de imediato 

sucesso 

Nprox:          INC R2 

                BR Nciclo 

Alínea C 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

                ; função TesteVetorN solicitada na alínea C 

                ; Entrada: R1 - valor máximo de N; R2 - vetor 

                ; Saída: vetor em R2 corretamente preenchido  

                ; Não utilizar R3 e R4 para poupar a cópia para o stack ao 

chamar TesteNM 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

TesteVetorN:    MOV R6, R0      ; R6=N de 0 a N(máximo)-1  

VNciclo:        CMP R6, R1       

                JMP.Z FimVN 

                CALL VNTeste    ; função com utilização de registos integrada 

nesta 

                                ; obtém o valor de N, e coloca em R5 

                MOV R4, R6      ; utilização temporária de R4, com a referência 

da posição em causa 

                ADD R4, R2 

                MOV M[R4], R5   ; guarda o resultado no vetor 

                INC R6 

                BR VNciclo 

                 

VNTeste:        CMP R6, 2       ; R6=N  

                JMP.NP VNsucesso  

                MOV R5, R0      ; R5=M, que vai de 0 a N-1 

VNciclo2:       CMP R5, R6 

                BR.Z VNfalha  

                                ; verificar se R5=M pertence a S 

                MOV R4, R5      ; utilização temporária de R4, com a referência 

da posição em causa 

                ADD R4, R2 

                CMP M[R4], R0 

                BR.Z VNprox     ; neste caso salta R5=M, já que não pertence a S 

                 

                ; chamar TesteNM com os valores atuais 

                PUSH R1         ; não é necessário guardar R3 e R4, podem ser 

destruidos por TesteNM 

                PUSH R2 

                MOV R1, R6 

                MOV R2, R5 

                CALL TesteNM    ; R1 fica com o resultado 

                CMP R1, R0 

                POP R2 

                POP R1 

                BR.NZ VNsucesso 

VNprox:         INC R5 

                BR VNciclo2 

VNfalha:        MOV R5, R0 

                RET 

VNsucesso:      MOV R5, 1 

FimVN:          RET 

Alínea D 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

                ; função TesteVetorB solicitada na alínea D 

                ; clone de TesteVetorN, mas índices substituídos 

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

                ; referência a uma posição N do vetor: 

                ; R2 - início - igual à função anterior 

                ; A - Número de posições; A=N>>4 

                ; B - bit ativo, inicia-se 1h, e vai-se rodando um bit para a 

esquerda  

                ;     (passa a 2h, 4h, ... até 8000h e de seguida de volta a 1h) 

                ;     tem um e um só bit ativo, o bit que se está a iterar 

                ; Operação de consulta: 

                ;       TEST M[A], B --- a flag Z ficará ativa se o bit em B 

estiver a zero 

                ; Operação de gravação para 1: 

                ;       OR M[A], B  

                ; Operação de gravação para 0: 

                ;       AND M[A], ~B  

                ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 

 

TesteVetorB:    MOV R6, R0      ; R6=N de 0 a N(máximo)-1, R7=B (bit em 

rotação)  

                MOV R7, 1       ; B iniciado na posição 0 

VBciclo:        CMP R6, R1       

                JMP.Z FimVN 

                CALL VBTeste    ; função com utilização de registos integrada 

nesta 

                                ; obtém o valor de N, e coloca em R5 

                ; Gravação 

                MOV R4, R6      ; R4=A=N>>4 

                SHR R4, 4       ; <<< única diferença para obter a posição de 

memória 

                ADD R4, R2 

                CMP R5, R0 

                BR.Z  VBGrava0   

                ; Gravação para 1 

                OR M[R4], R7    ; operação de gravação utiliza R7=B 

                BR VBprox 

VBGrava0:       COM R7 

                AND M[R4], R7   ; operação de gravação utiliza ~B 

                COM R7 

VBprox:         ROL R7, 1       ; o bit vai sempre movendo, ao mesmo tempo que N 

é incrementado  

                INC R6          ; R6 é incrementado de igual forma, tem o valor 

de N como na função anterior 

                BR VBciclo 

                 

VBTeste:        PUSH R7         ; R7 é também B mas deste ciclo, pelo que há que 

guardar o anterior 

                CMP R6, 2       ; R6=N  

                JMP.NP VBsucesso  

                MOV R5, R0      ; R5=M, que vai de 0 a M-1 

                MOV R7, 1       ; B iniciado na posição 0 

VBciclo2:       CMP R5, R6 

                BR.Z VBfalha  

                ; verificar se R5=M pertence a S 

                MOV R4, R5      ; R4=A=N>>4 

                SHR R4, 4 

                ADD R4, R2 

                TEST M[R4], R7 

                BR.Z VBprox2    ; neste caso salta R5=M, já que não pertence a S 

                 

                ; chamar TesteNM com os valores atuais, igual à função anterior 

                PUSH R1 

                PUSH R2 

                MOV R1, R6 

                MOV R2, R5 

                CALL TesteNM  

                CMP R1, R0 

                POP R2 

                POP R1 

                BR.NZ VBsucesso 

VBprox2:        ROL R7, 1       ; não esquecer de rodar o bit, sempre que o 

correspondente valor de N é incrementado 

                INC R5 

                BR VBciclo2 

VBfalha:        MOV R5, R0 

                POP R7 

                RET 

VBsucesso:      MOV R5, 1 

                POP R7 

FimVB:          RET 

 

Qualquer questão coloquem, tanto relativo a esta resolução exemplo, como relativo à vossa nota, 

tendo em atenção o avaliador, é ele quem pode rever a vossa nota se for o caso. 

Cumprimentos, 

José Coelho 

 


